Example of DAE (index reduction)

```
> with(plots):
```

The DAE (pendulum in cartesian coordinate)

```
> EQ1 := diff(x(t),t)-u(t) ;
  EQ2 := diff(y(t),t)-v(t) ;
  EQ3 := m*diff(u(t),t)+x(t)*lambda(t) ;
  EQ4 := m*diff(v(t),t)+y(t)*lambda(t)+m*g ;
  EQ5 := x(t)^2+y(t)^2-1 ;
```

$$EQ1 := \frac{d}{dt} x(t) - u(t)$$

$$EQ2 := \frac{d}{dt} y(t) - v(t)$$

$$EQ3 := m \left(\frac{d}{dt} u(t) \right) + x(t)\, \lambda(t)$$

$$EQ4 := m \left(\frac{d}{dt} v(t) \right) + y(t)\, \lambda(t) + m\, g$$

$$EQ5 := x(t)^2 + y(t)^2 - 1 \tag{1}$$

```
> SUBS_RULE := solve( {EQ||(1..4)}, diff({x(t),y(t),u(t),v(t)},t) ) ;
```

$$SUBS\_RULE := \left\{ \frac{d}{dt} u(t) = -\frac{x(t)\, \lambda(t)}{m}, \frac{d}{dt} v(t) = -\frac{y(t)\, \lambda(t) + m\, g}{m}, \frac{d}{dt} x(t) = u(t), \frac{d}{dt} y(t) = v(t) \right\} \tag{2}$$

Step 1, derivation of the "algebraic part"

```
> DEQ5 := diff(EQ5,t) ; DEQ5 := subs( SUBS_RULE, DEQ5) ;
```

$$DEQ5 := 2\, x(t) \left(\frac{d}{dt} x(t) \right) + 2\, y(t) \left(\frac{d}{dt} y(t) \right)$$

$$DEQ5 := 2\, x(t)\, u(t) + 2\, y(t)\, v(t) \tag{3}$$

Do not contain derivative of lambda(t), need another derivation

```
> DDEQ5 := diff(DEQ5,t) ;
  DDEQ5 := simplify(subs( SUBS_RULE, DDEQ5)) ;
  DDEQ5 := simplify(subs( x(t)^2=1-y(t)^2, DDEQ5)) ;
```

$$DDEQ5 := 2 \left(\frac{d}{dt} x(t) \right) u(t) + 2\, x(t) \left(\frac{d}{dt} u(t) \right) + 2 \left(\frac{d}{dt} y(t) \right) v(t) + 2\, y(t) \left(\frac{d}{dt} v(t) \right)$$

$$DDEQ5 := \frac{2 \left(u(t)^2\, m + v(t)^2\, m - x(t)^2\, \lambda(t) - \lambda(t)\, y(t)^2 - y(t)\, g\, m \right)}{m}$$

$$DDEQ5 := \frac{2 \left(u(t)^2\, m + v(t)^2\, m - y(t)\, g\, m - \lambda(t) \right)}{m} \tag{4}$$

```
> EQLAMBDA := solve(DDEQ5,{lambda(t)} ) ;
```

$$EQLAMBDA := \left\{ \lambda(t) = m \left(u(t)^2 + v(t)^2 - y(t)\, g \right) \right\} \tag{5}$$

Do not contain derivative of lambda(t), need another derivation

```
> DDDEQ5 := diff(DDEQ5,t) ;
```

```
DDDEQ5 := simplify(subs( SUBS_RULE, DDDEQ5)) ;
```

$$DDDEQ5 := \frac{2\left(2\,u(t)\,m\left(\frac{\mathrm{d}}{\mathrm{d}t}u(t)\right)+2\,v(t)\,m\left(\frac{\mathrm{d}}{\mathrm{d}t}v(t)\right)-\left(\frac{\mathrm{d}}{\mathrm{d}t}y(t)\right)g\,m-\left(\frac{\mathrm{d}}{\mathrm{d}t}\lambda(t)\right)\right)}{m}$$

$$DDDEQ5 := -\frac{2\left(2\,u(t)\,x(t)\,\lambda(t)+2\,\lambda(t)\,v(t)\,y(t)+3\,v(t)\,g\,m+\frac{\mathrm{d}}{\mathrm{d}t}\lambda(t)\right)}{m} \quad (6)$$

Solve rispect to the derivative of lambda(t)

```
> solve( DDDEQ5, {diff(lambda(t),t)} ) ; SOL5 := op(collect( %,
  lambda)) ;
```

$$\left\{\frac{\mathrm{d}}{\mathrm{d}t}\lambda(t) = -2\,u(t)\,x(t)\,\lambda(t)-2\,\lambda(t)\,v(t)\,y(t)-3\,v(t)\,g\,m\right\}$$

$$SOL5 := \frac{\mathrm{d}}{\mathrm{d}t}\lambda(t) = \left(-2\,x(t)\,u(t)-2\,y(t)\,v(t)\right)\lambda(t)-3\,v(t)\,g\,m \quad (7)$$

The ODE resulting from index reduction

```
> ODE := EQ||(1..4), SOL5 ;
```

$$ODE := \frac{\mathrm{d}}{\mathrm{d}t}x(t)-u(t),\ \frac{\mathrm{d}}{\mathrm{d}t}y(t)-v(t),\ m\left(\frac{\mathrm{d}}{\mathrm{d}t}u(t)\right)+x(t)\,\lambda(t),\ m\left(\frac{\mathrm{d}}{\mathrm{d}t}v(t)\right)+y(t)\,\lambda(t)+m\,g,\ \frac{\mathrm{d}}{\mathrm{d}t}\lambda(t) = \left(-2\,x(t)\,u(t)-2\,y(t)\,v(t)\right)\lambda(t)-3\,v(t)\,g\,m \quad (8)$$

```
> INI := x(0)=1,y(0)=0,u(0)=0,v(0)=0 ;
  INI := INI, subs(INI,lambda(0)=subs(t=0,subs(EQLAMBDA,lambda(t))))
  ;
```

$$INI := x(0)=1,\ y(0)=0,\ u(0)=0,\ v(0)=0$$

$$INI := x(0)=1,\ y(0)=0,\ u(0)=0,\ v(0)=0,\ \lambda(0)=0 \quad (9)$$

```
> SUBS_VALUES := g=9.81, m=1 ;
```

$$SUBS\_VALUES := g=9.81,\ m=1 \quad (10)$$

```
> SOL := dsolve( subs(SUBS_VALUES,{ODE,INI}),type=numeric, range=0.
  .40 ) ;
```

$SOL :=$ **proc**(x_rkf45) ... **end proc** (11)

```
> odeplot(SOL, [x(t),y(t)], 0..38, numpoints=1000) ;
```




```
> INI := x(0)=1,y(0)=0,u(0)=0,v(0)=0 ;
```

$$INI := x(0) = 1, y(0) = 0, u(0) = 0, v(0) = 0 \tag{12}$$

```
> DAE := EQ||(1..5) ;
```

$$DAE := \frac{d}{dt} x(t) - u(t), \frac{d}{dt} y(t) - v(t), m \left(\frac{d}{dt} u(t)\right) + x(t)\,\lambda(t), m\left(\frac{d}{dt} v(t)\right) + y(t)\,\lambda(t) + m\,g, x(t)^2 + y(t)^2 - 1 \tag{13}$$

```
> SOL_DAE := dsolve( subs(SUBS_VALUES,{DAE,INI}),type=numeric ) ;
```

$SOL\_DAE :=$ **proc**(x_rkf45_dae) ... **end proc** **(14)**

```
> odeplot(SOL_DAE, [x(t),y(t)], 0..54,numpoints=1000) ;
```




```
> example(odeplot) ;
```